Splunk: transactionコマンドについて


###### 実施環境： Splunk Free 8.2.2

### 0. 概要

ログによっては、例えば以下のように複数のログが一連のイベントを表す場合があります。

10:21 ユーザ001 ログイン
10:23 ユーザ001 商品Aをかごに入れる
10:24 ユーザ001 商品Bをかごに入れる
10:28 ユーザ001 商品Aをかごから出す
10:29 ユーザ001 商品Cをかごに入れる
10:32 ユーザ001 購入
10:33 ユーザ001 ログアウト

このようなログについて、同じイベントを表すログを1つにまとめて解析したいことがあります。
そのようなときに用いるのが、 transaction コマンドです。
このコマンドは時間やフィールド値によって同じイベントを表すログをグループ化し、1つにまとめることができるコマンドです。
今回はこの transaction コマンドについて紹介します。

### 1. 基本

今回は以下のようなログを自作して使用しました。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| sort _time
| table _time, _raw, host_name, user_name, action
```

![WS000429.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/9a5d0b7d-7f95-bfc6-de25-cb5de3802090.jpeg "元ログ")

 transaction コマンドの基本的な文法は以下の通りです。
キーとなるフィールドを指定して、その値をもとにグループ化します。

 transaction フィールド名1, フィールド名2, ...

試しに、「 user_name 」を指定して transaction コマンドを実行してみます。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| transaction user_name
| sort _time
| table _time, _raw, host_name, user_name, action, closed_txn, duration, eventcount, field_match_sum
```

![WS000435.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/31f24041-96e3-b5a0-0a48-e897710726a2.jpeg "キー1個")

「 user_name 」が「 user1 」のデータと「 user2 」のデータで、2つにグループ化されました。

次に、「 host_name 」も追加で指定してみます。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| transaction user_name, host_name
| sort _time
| table _time, _raw, host_name, user_name, action, closed_txn, duration, eventcount, field_match_sum
```

![WS000436.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/85c6c3d8-8c37-e109-975f-6d8c551076c0.jpeg "キー2個")

グループ化の条件に、「 host_name 」も追加されました。

### 2. 追加されるフィールド

さて、上記の SPL で、 table コマンドに指定しているフィールドが transaction コマンドを入れる前後で増えていることにお気付きでしょうか。
これは単に指定を増やしたのではなく、実際に transaction コマンドを通すとフィールドがいくつか増えるのです。
実際に詳細モードで SPL を実行して確認してみます。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
```

![WS000430.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/b477e58f-cc5a-7326-58fa-da386b8bebde.jpeg "transactionなし")

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| transaction user_name, host_name
```

![WS000431.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/db1f2e43-43f1-1682-1b67-277ba1b4a4ee.jpeg "transactionあり")

「 closed_txn 」「 duration 」「 eventcount 」「 field_match_sum 」の4つのフィールドが追加されていることがわかります。
「 closed_txn 」は基本的に0ですが、後述の「 maxevents 」「 maxpause 」「 maxspan 」「 startswith 」のいずれかを指定し、それが適用された場合に1になります。
「 duration 」はそのデータにまとめられたログの、最初のログから最後のログまでの秒数です。今回は8:19:01から8:20:22までの81秒です。
「 eventcount 」はそのデータにまとめられたログの件数です。
「 field_match_sum 」はそのデータにまとめる際にヒットしたフィールドの個数です。今回は「 user_name 」「 host_name 」の2つのフィールドをコマンドで指定しているので、2×2=4個となります。

### 3. オプション

前述の SPL ではログをグループ化するのにフィールドを使用しました。
さらに transaction コマンドのオプションによって、時間やログデータの文字列をグループ化に使用することができます。

 maxpause を指定した場合、ログとログの間の時間が指定した時間より大きいとその2つのログは別のグループとして扱います。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| transaction maxpause=5m user_name, host_name
| sort _time
| table _time, _raw, host_name, user_name, action, closed_txn, duration, eventcount, field_match_sum
```

![WS000437.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/501e0225-5f46-1cdb-b699-6d978624b871.jpeg "maxpause")

 maxspan を指定した場合、グループの最初のログと特定のログの間の時間が指定した時間より大きいとその2つのログは別のグループとして扱います。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| transaction maxspan=30m user_name, host_name
| sort _time
| table _time, _raw, host_name, user_name, action, closed_txn, duration, eventcount, field_match_sum
```

![WS000438.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/1b27e1e6-e75f-735b-fb57-829c30c7855f.jpeg "maxspan")

 startswith を指定した場合、指定した文字列が含まれるログでグループが区切られ、該当のログはグループの最初のログとして扱われます。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| transaction startswith="login" user_name, host_name
| sort _time
| table _time, _raw, host_name, user_name, action, closed_txn, duration, eventcount, field_match_sum
```

![WS000439.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/b7ffee68-2549-73b6-23f7-85846a6a345d.jpeg "startswith")

 endswith を指定した場合、指定した文字列が含まれるログでグループが区切られ、該当のログはグループの最後のログとして扱われます。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| transaction endswith="logout" user_name, host_name
| sort _time
| table _time, _raw, host_name, user_name, action, closed_txn, duration, eventcount, field_match_sum
```

![WS000440.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/cdd4b62e-2b42-f221-8d9d-47bbc7fc44d2.jpeg "endswith")

### 4. stats コマンドでの代用

ここまで transaction コマンドについて紹介してきましたが、実のところ transaction コマンドは性能面等から見てあまり良いコマンドではなかったりします。
そのため、可能であれば以下のように stats コマンド等で代用することが好ましいです。

```Splunk:Splunk
index="test_index" sourcetype="test_access"
| eval time = _time
| bin span=30m _time
| stats count AS eventcount, min(time) AS min_time, max(time) AS max_time BY _time, host_name, user_name
| eval _time = min_time
| eval duration = max_time - min_time
| sort _time
| table _time, host_name, user_name, duration, eventcount
```

![WS000441.JPG](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/1465018/c1327425-6805-7532-96ef-843b7d9df43e.jpeg "stats")

 transaction コマンドが必要となる場面は主に、フィールド値だけではログのグループ化が難しい場面(なおこの場合、前述のオプションが必要となることが多い)です。
そうでない場合は stats コマンドでの代用を検討することをおすすめします。